


Lehrbeispiel 2.1: Kennlinie einer Glühlampe



Bibliotheken aus [2a]

DC-Analyse
 DC-Main-Sweep

Bild 2.7: Simulation einer Glühlampe

* Schematics Netlist LB_2.1 *

```
V_Uq      1      0      1V
G_G1      1      0      VALUE { 100mA*PWR(V(1, 0)/1V,0.3) }
```

** Analysis setup **

```
.DC      LIN      V_Uq 0      10      0.01
.TEMP    20
.OP
```

*** RESUMING LB_2_1.cir ***


```
.probe
.END
```

- Aufruf von Variablen und Darstellung von Funktionen im **PROBE-Fenster** über:

Trace → *Add Trace* I(G_G1) [Kennlinie der Glühlampe $I = f(U)$]

Weitere Informationen (ohne Schaltung): *Analysis* → *Examine Output* (Output-File)

Lehrbeispiel 2.3: Temperaturabhängigkeit



Bibliotheken aus [2a]

DC-Analyse
DC-Main-Sweep
DC-Nested-Sweep

Bild 2.11: Schaltung für den Temperatur-Sweep

* Schematics Netlist LB_2.3 *

```
V_Uq      1      0      10V
R_R1      1      0      Rbreak    1k
```

** Analysis setup **

```
.DC          LIN  TEMP -50  100  0.1
+ Res       Rbreak(TC1) LIST -0.6m -0.03m 0.2m
.OP
```

*** RESUMING LB_2_3.cir ***

```
.probe
.END
```

- Aufruf von Variablen und Darstellung von Funktionen im **PROBE-Fenster** über:


Trace → *Add Trace* V(1)/I(R_R1) [Widerstandsänderung bei Temperaturänderung]

Weitere Informationen: *Analysis* → *Examine Output* (Output-File)



Lehrbeispiel 2.4: Simulation SKIN-Effekt

(NEU: 2020)



Bibliotheken aus [2a]

AC-Analyse
 AC-Sweep (Decade)

^ Bild 2.14: Ohmscher Widerstand (HF)

* Schematics Netlist LB_2.4 *

```
V_Uq      1      0      AC      1V      SIN 0 1V 1kHz 0 0 0
R_RSkin   1      0      1
```

** Analysis setup **

```
.AC DEC 1000 10kHz 100Meg
.TEMP 20
.OP
```

*** RESUMING LB_2_4.cir ***

```
.probe
.END
```

- Aufruf von Variablen und Darstellung von Funktionen im **PROBE-Fenster** über:


Trace → *Add Trace* V(1) ändern in:
 0.5m*SQRT(2*FREQUENCY) [Frequenzgang der Hilfsvariablen *x*]

Weitere Informationen: *Analysis* → *Examine Output* (Output-File)



Lehrbeispiel 2.5: HF-Ersatzschaltbild Kondensator

(Orig.: LB_2.4)



Bibliotheken aus [2a]

AC-Analyse
 AC- Sweep (Decade)

Bild 2.20: Simulation der HF-Ersatzschaltung

* Schematics Netlist LB_2.5 *

```
V_Uq      ein  0  AC  1V  SIN  0  1V  1k  0  0  0
L_Lers    ein  2  20nH
R_R_ESR   2    3  0.1
C_C       3    0  100nF
```

** Analysis setup **

```
.AC      DEC  10000 10k  100MEG
.TEMP    20
.OP
```

*** RESUMING LB_2_5.cir ***

```
.probe
.END
```


- Aufruf von Variablen und Darstellung von Funktionen im **PROBE-Fenster** über:

Trace → *Add Trace* $V(\text{ein})/I(R_R_ESR)$ [Frequenzgang der Impedanz]
 ändern: *Plot* → *Axis Settings* → *Y-Axis*: • Log

Weitere Informationen: *Analysis* → *Examine Output* (Output-File)

Lehrbeispiel 2.6: Wicklungskapazität einer Spule

(Orig.: LB_2.5)



Bibliotheken aus [2a]
 AC-Analyse
 AC-Sweep (Decade)
 Parametric-Sweep

Bild 2.31: Schaltung zum Lehrbeispiel 2.6

* Schematics Netlist LB_2.6 *

```
I_Iq      0      1      AC      10mA SIN      0      10mA 1kHz 0      0      0
R_RFe     2      0      100k
R_RCu     1      2      20
C_C1      1      0      {Cers}
C_Cw      2      0      10p
L_L1      2      0      10mH
.PARAM    Cers=10pF
```

** Analysis setup **

```
.AC      DEC      1000      100k      1MEG
.STEP    PARAM    Cers      LIST      0      20p      50p      100p
.TEMP    20
.OP
```

*** RESUMING LB_2_6.cir ***

```
.probe
.END
```


- Aufruf von Variablen und Darstellung von Funktionen im **PROBE-Fenster** über:

Trace → Add Trace V(1) [Frequenzgänge der Spannung U_{Spule}]

Weitere Informationen: Analysis → Examine Output (Output-File)

Lehrbeispiel 2.7: Magnetisierungskennlinie

(Orig.: LB_2.6)



Bibliotheken aus [2a]

DC-Analyse
 DC-Main-Sweep

Bild 2.35: Schaltung zur Aufnahme der Magnetisierungskennlinie

* Schematics Netlist LB_2.7 *

```
I_Iq      0      P      DC      1A
L1_TX1    P      0      10H
L2_TX1    S      0      0.1H
K_TX1     L1_TX1  L2_TX1    1      kbreak
R_R1      S      0      10MEG
```

** Analysis setup **

```
.DC LIN    I_Iq  0A    10A    1mA
.TEMP      20
.OP
```

*** RESUMING LB_2.7.cir ***

```
.probe
.END
```

- Aufruf von Variablen und Darstellung von Funktionen im **PROBE-Fenster** über:


Trace → *Add Trace* B(K_TX1) [magnetische Flussdichte]

Anmerkung: Die Achsen müssen noch nachbearbeitet werden (Einheiten nach DIN).

Weitere Informationen: *Analysis* → *Examine Output* (Output-File)



Lehrbeispiel 2.8: Kennlinie Heißleiter (Orig.: LB_2.7)



Bibliotheken aus [2a]

DC-Analyse
 DC-Main-Sweep

Bild 2.51: Schaltung zum Lehrbeispiel 2.8

* Schematics Netlist LB_2.8 *

```
V_Uq      1      0      10V
R_R1      1      0      Rbreak {A*EXP(B/T)}
.PARAM    A=21.3158m B=3420
.PARAM    T=293 TC1=-40m

** Analysis setup **
.DC       LIN   PARAM      T      233   393   1
.TEMP     20
.OP


*** RESUMING LB_2_8.cir ***
.probe
.END
```

- Aufruf von Variablen und Darstellung von Funktionen im **PROBE-Fenster** über:

Trace → *Add Trace* V(1)/I(R_R1) [Widerstands-Temperatur-Kennlinie]

Weitere Informationen: *Analysis* → *Examine Output* (Output-File)

Lehrbeispiel 2.9: Kennlinie Kaltleiter (Orig.: LB_2.8)



Bibliotheken aus [2a]
 DC-Analyse
 DC-Main-Sweep

Bild 2.57: Schaltung zum Lehrbeispiel 2.9

* Schematics Netlist LB_2.9 *

```
V_Uq      ein  0   1V
R_Rmin    ein  1   150
R_Rmax    1    0   220k
R_RT      1    0   {0.1*EXP(T*3000)}
.PARAM    T=50
```

** Analysis setup **

```
.DC      LIN  PARAM    T    20    300    1
.TEMP    20
.OP
```

*** RESUMING LB_2_9.cir ***

```
.probe
.END
```


- Aufruf von Variablen und Darstellung von Funktionen im **PROBE-Fenster** über:

Trace → *Add Trace* V(ein)/I(R_Rmin) [Widerstands-Temperatur-Kennlinie]

Weitere Informationen: *Analysis* → *Examine Output* (Output-File)

Lehrbeispiel 2.11: Kennlinien Fotowiderstand

(Orig.: LB_2.10)



Bibliotheken aus [2a]

DC-Analyse
 DC-Main-Sweep (Decade)

Bild 2.65: Schaltung zum Lehrbeispiel 2.11

* Schematics Netlist LB_2.11 *

```
V_Uq      1      0      1V
R_Rp      1      0      {RpH*PWR(E,-GAM)}
.PARAM    RpH=149.5k  GAM=0.825
.PARAM    E=1_lx
```

** Analysis setup **

```
.DC      DEC  PARAM      E      0.1  1000  1000
.TEMP    20
.OP
```

*** RESUMING LB_2.11.cir ***

```
.probe
.END
```

- Aufruf von Variablen und Darstellung von Funktionen im **PROBE-Fenster** über:

Trace → *Add Trace* V(1)/I(R_Rp) [Kennlinie: $R_p = f(E)$]

Weitere Informationen: *Analysis* → *Examine Output* (Output-File)



Lehrbeispiel 2.13: Kennlinie der Universaldiode 1N 4148

(Orig.: LB_2.12)

Tabelle 2.8: Modell-Parameter von EVAL-Dioden

Parameter	1N 4002	1N 4148	1N 750
IS in A	14.11 n	2.682 n	0.8805 f
RS in Ω	33.89 m	566.4 m	250 m
N	1.984	1.836	1
CJO in F	51.17 p	4 p	175 p
TT in s	4.761 μ	11.54 n	0.1 n

Bibliotheken aus [2a]

 DC-Analyse
 DC-Main-Sweep
 Variation von Modell-Parametern

* Schematics Netlist LB_2.13 *

```
V1      1      0      DC      1
D1      1      0      DIODE1
D2      1      0      DIODE2
D3      1      0      DIODE3
.MODEL  DIODE1  D      (IS=2.682n N=1.836 RS=0.5664 CJO=4p TT=11.54n)
.MODEL  DIODE2  D      (IS=5.364n N=1.836 RS=0.5664 CJO=4p TT=11.54n)
.MODEL  DIODE3  D      (IS=1.341n N=1.836 RS=0.5664 CJO=4p TT=11.54n)
```

```
** Analysis setup **
.DC      V1      0.5      1      1m
.TEMP    20
.OP
```

```
*** RESUMING LB_2_13.cir ***
.probe
.END
```


- Aufruf von Variablen und Darstellung von Funktionen im **PROBE-Fenster** über:

Trace → *Add Trace* I(D1) I(D2) I(D3) [Strom-Spannungs-Kennlinie (I_S variabel)]

Weitere Informationen: *Analysis* → *Examine Output* (Output-File)

Lehrbeispiel 2.14: Zweiweggleichrichtung

(Orig.: LB_2.13)



Bibliotheken aus [2a]
 Transienten-Analyse
 (in der Netzliste zur Übung
 mit **Sw_tClose**)

Bild 2.108: Zweiweggleichrichtung

* Schematics Netlist LB_2.14 *

```
V_Up      2      0      SIN  0      15V  50Hz  0      0      0
V_Un      0      3      SIN  0      15V  50Hz  0      0      0
D_V1      2      1      D1N4002
D_V2      3      1      D1N4002
X_U1      1      4      Sw_tClose
+ PARAMS: tClose=60ms ttran=1us Rclosed=1m Ropen=1G
R_Ra      1      0      100
C_Ca      4      0      470uF
```

** Analysis setup **

```
.TRAN      0      120ms 0      10us
.TEMP      20
.OP
```

*** RESUMING LB_2_14.cir ***

```
.probe
.END
```


- Aufruf von Variablen und Darstellung von Funktionen im **PROBE-Fenster** über:

Trace → Add Trace V(2) V(3) und: V(1) [Zeitfunktionen u_p und u_n sowie u_a]

Weitere Informationen: Analysis → Examine Output (Output-File)

Lehrbeispiel 2.15: Schaltverhalten von Dioden

(Orig.: LB_2.14)



Bibliotheken aus [2a]
 Transienten-Analyse

Bild 2.112: Schaltung zum Lehrbeispiel 2.15

* Schematics Netlist LB_2.15 *

```
V_UP      1      0      DC      0      AC      0      PULSE      -4V      1V      0      0
          0      0.5ns  1ns
D_V1      1      2      MBD101
D_V2      1      3      D1N4148
D_V3      1      4      D1N4002
R_Rv1     2      0      10
R_Rv2     3      0      10
R_Rv3     4      0      10
```

** Analysis setup **

```
.TRAN      0      1.5ns  0      0.1ps
.TEMP      20
.OP
```

*** RESUMING LB_2_15.cir ***

```
.probe
.END
```


- Aufruf von Variablen und Darstellung von Funktionen im **PROBE-Fenster** über:

Trace → *Add Trace* I(D_V1) I(D_V2) I(D_V3) [Stromverläufe i_{Dx}]

Weitere Informationen: *Analysis* → *Examine Output* (Output-File)

Lehrbeispiel 2.16.a: ODER-Gatter

(Orig.: LB_2.15)



Bibliotheken aus [2a]
 Transienten-Analyse

Bild 2.119: Simulation eines ODER-Gatters

* Schematics Netlist LB_2.16.a (ODER) *

```
V_Ux1      x1    0    DC    0    AC    0    PULSE 0V 5V 0 10n 10n 5ms 10ms
V_Ux2      x2    0    DC    0    AC    0    PULSE 0V 5V 0 10n 10n 4ms 12ms
X_VD3      x1    x2    y    Dbreak3
R_Ra       y     0     1k
```

** Analysis setup **

```
.TRAN      0     40ms 0     1ms
.TEMP      20
.OP
```

*** RESUMING LB_2_16_a.cir ***

```
.probe
.END
```

- Aufruf von Variablen und Darstellung von Funktionen im **PROBE-Fenster** über:


Trace → Add Trace V(x1) V(x2) V(y) [Zeitfunktionen $u_{x1}/u_{x2}/u_y$]

Weitere Informationen: *Analysis* → *Examine Output* (Output-File)



Lehrbeispiel 2.16.b: UND-Gatter

(Orig.: LB_2.15)



Bibliotheken aus [2a]

Transienten-Analyse

Bild 2.121: Simulation eines UND-Gatters

* Schematics Netlist LB_2.16.b (UND) *

```
V_UB      1      0      5V
V_Ux1     x1     0      DC    0      AC    0      PULSE 0V 5V 0 10n 10n 5ms 10ms
V_Ux2     x2     0      DC    0      AC    0      PULSE 0V 5V 0 10n 10n 4ms 12ms
D_V1      y      x1     D1N4148
D_V2      y      x2     D1N4148
R_R1      x1     0      100
R_R2      x2     0      100
R_RU      1      y      50k
```

** Analysis setup **

```
.TRAN      0      40ms 0      0.3ms
.TEMP      20
.OP
```

*** RESUMING LB_2_16_b.cir ***

```
.probe
.END
```


- Aufruf von Variablen und Darstellung von Funktionen im **PROBE-Fenster** über:

Trace → Add Trace V(x1) V(x2) V(y) [Zeitfunktionen $u_{x1}/u_{x2}/u_y$]

Weitere Informationen: *Analysis* → *Examine Output* (Output-File)

Lehrbeispiel 2.17: Kennlinie einer Z-Diode

(Orig.: LB_2.16)



Bibliotheken aus [2a]

DC-Analyse / DC-Main-Sweep
 Variation von Modell-Parametern

Bild 2.125: Simulation der Z-Diode 1N 750

* Schematics Netlist LB_2.17.a (Kennlinie) *

```
V_Uq      1      0      -3V
D_V1      1      0      D1N750
** Analysis setup **
.DC       LIN   V_Uq  0 -5 1m
.TEMP     20
.OP
*** RESUMING LB_2_17_a.cir ***
.probe
.END
```

• Aufruf von Variablen / Darstellung
 im **PROBE-Fenster** über:

- Trace → Add Trace I(D_V1)
- Trace → Add Trace
 I(D1) I(D2) I(D3) I(D41)

Weitere Informationen:


Analysis → Examine Output
 (Output-File)

* Schematics Netlist LB_2.17.b (Modell-Parameter) *

```
V1        1      0      DC    0
D1        1      0      DIODE1
D2        1      0      DIODE2
D3        1      0      DIODE3
D4        1      0      DIODE4
.MODEL    DIODE1    D    (Is=880.5E-18 Rs=.25 Bv=3.3 Ibv=20.245m Nbv=1.7)
.MODEL    DIODE2    D    (Is=880.5E-18 Rs=.25 Bv=4.7 Ibv=20.245m Nbv=1.7)
.MODEL    DIODE3    D    (Is=880.5E-18 Rs=.25 Bv=6.2 Ibv=20.245m Nbv=1.7)
.MODEL    DIODE4    D    (Is=880.5E-18 Rs=.25 Bv=7.5 Ibv=20.245m Nbv=1.7)
** Analysis setup **
.DC       V1     -8     0     1m
.TEMP     20
.OP
*** RESUMING LB_2_17_b.cir ***
.probe
.END
```

Lehrbeispiel 2.18: Kapazitätsdiode

(Orig.: LB_2.17)



Bibliotheken aus [2a]

AC-Analyse
 AC-Sweep
 Parametric-Sweep

Bild 2.133: Simulation einer Kapazitätsdiode

* Schematics Netlist LB_2.18 *

```
V_Uq      a      0      AC   1V      SIN   0      1V   1kHz  0      0      0
L_L1      a      b      10mH
D_V1      b      c      MV2201
R_Ri      c      0      50
```

** Analysis setup **

```
.AC          LIN   50000 380k  440k
.STEP       TEMP LIST  -50  -10  20   60   100
.OP
```

*** RESUMING LB_2_18.cir ***

```
.probe
.END
```

- Aufruf von Variablen und Darstellung von Funktionen im **PROBE-Fenster** über:


Trace → *Add Trace* I(D_V1) [Amplitudenfrequenzgänge des Stromes]

Weitere Informationen: *Analysis* → *Examine Output* (Output-File)



Lehrbeispiel 2.19: SCHOTTKY-Diode

(Orig.: LB_2.18)



Bibliotheken aus [2a]

DC-Analyse
DC-Main-Sweep

Bild 2.140: Simulation von Durchlasskennlinien

* Schematics Netlist LB_2.19 *

```
V_Uq      1      0      1V
D_V1      1      0      MBD101
D_V2      1      0      D1N4002
```

** Analysis setup **

```
.DC      LIN  V_Uq 0      1      1m
.TEMP    20
.OP
```

*** RESUMING LB_2_19.cir ***

```
.probe
.END
```

- Aufruf von Variablen und Darstellung von Funktionen im **PROBE-Fenster** über:

Trace → *Add Trace* I(D_V1) I(D_V2) [Strom-Spannungs-Kennlinien der Dioden]

Weitere Informationen: *Analysis* → *Examine Output* (Output-File)